Paper 1873-2014

Modeling Ordinal Responses for a Better Understanding of Drivers of Customer Satisfaction

Jorge Alejandro, Market Probe; Sharon Kim, Market Probe

ABSTRACT

While survey researchers make great attempts to standardize their questionnaires – including the usage of ratings scales – oftentimes the methods applied rely on the assumption that the collected response is interval level, as opposed to ordinal. Although analytical approaches that are better equipped to handle ordinal responses have become more accessible over the last several years, many times their use is affected by the requirements of the method, including sample size. Furthermore, often there are sub-populations of interest that are left unexplored because of the additional strain this would impose on the sample size requirement. This paper describes an implementation that uses a Bayesian approach to model an ordinal response while capturing the distribution of heterogeneity in the included sub-populations. More specifically, SAS® PROC MCMC is used to carry out a comprehensive modeling strategy of ratings data that accounts for the use of an ordinal response scale.

INTRODUCTION

Survey researchers are commonly charged with the task of identifying key drivers of attitudinal outcomes such as overall evaluations or intentions (e.g., overall customer satisfaction, likelihood to recommend a product/service) In doing so, respondents of these surveys are asked to place themselves on a k-point scale in order to reflect the intensity of their opinion. When these types of responses are used to derive key drivers, typically the analytical approach of choice treats these responses as if they were interval scale in a regression-type setting, where the effect size is used to quantify the magnitude of the drivers. While attempts are made with the purpose of justifying the application of these types of methodologies, alternative modeling strategies that recognize the ordinal aspect of the response aren't as prevalent in applied survey research.

In this paper, a cumulative logit model is employed to model the ordered attitudinal response as a function of other survey items; PROC MCMC is used to carry out the modeling strategy. Additionally, sub-populations present in the data are recognized and incorporated into the model as a multivariate random-effect in order to better capture the group-specific variability.

ORDINAL RESPONSE IN SURVEY RESEARCH

In social science, it is commonly accepted that the observed response Y is assumed to be the result of discrete realizations of an underlying continuous latent response Z. Since Z is unobserved, it is associated to Y by a set of "cut points" whose distances are unknown. For example, in the case of a 5-point scale survey response, these cut-points are such that:

$$Y=1; \quad Z \le \alpha_1$$

$$Y=2; \quad \alpha_1 < Z \le \alpha_2$$

$$Y=3; \quad \alpha_2 < Z \le \alpha_3$$

$$Y=4; \quad \alpha_3 < Z \le \alpha_4$$

$$Y=5; \quad Z > \alpha_4$$

These cut-points divide the unobserved response Z in J categories and are used to relate the observed response with the covariates via the following expression, also known as a cumulative logit:

$$\log\left[\frac{\Pr(Y \le y_j | x)}{\Pr(Y > y_j | x)}\right] = \alpha_j - \boldsymbol{x}'\beta; \quad j = 1, 2, \ldots j-1 \qquad (1)$$

In this parameterization, α_j is the log-odds of falling into or below category j when x=0. A positive slope (beta coefficient) indicates a tendency for Y to increase when x increases.

GROUPING OF SCALE ITEMS

It is important to highlight that end-users of this type of research tend to prefer to work with a very specific grouping of scale items. Typically, these groupings are associated with the objectives behind these types of customer programs. For example, there are organizations concerned with tracking overall customer satisfaction (OSAT) and the percent of respondents that give them a rating of 9 or 10 on a 10-point scale (top-2 box.) In addition to this metric, managers are also interested in keeping track of the percent of respondents that give low evaluations: bottom-6 score.

An illustration of how scale items may be grouped is presented in Table 1 below. In this table, 'Attribute-n' corresponds to an attitudinal item also measured in the survey instrument that refers to an element of the customer experience that is believed to affect OSAT.

	Overall Satisfaction		
Attribute-n	**Bottom-6 (1-6)**	**Middle-2 (7,8)**	**Top-2 (9,10)**
Bottom-8 (1-8)			
Top-2 (9,10)			

Table 1. Grouping of scale items and **c**ross-tabulation between Attribute-n and Overall Satisfaction (OSAT).

This grouping of scale items is applied throughout the rest of this paper and is used to support the operationalization of the outcome obtained from this type of model: understand the impact a top-2 box score (vs. a bottom-8) can have on the ordered response. Also, the use of this grouping helps to relieve the sample size requirements that result when multiple attributes are included in the model (since this results in an increasing combination of the corresponding levels.)

CASE STUDY

In 2013, as part of a syndicated study, data was collected for several casual dining establishments in the US. This was an online study that was fielded over a one month period. Patrons of these establishments were asked to evaluate their overall level of satisfaction with their experience. Additionally, respondents also evaluated the restaurant on the following experience-related attributes:

- Variety of menu items
- Hostess service
- Friendly employees in restaurant
- Restaurant ambiance
- Wait time for getting seated
- Convenient location

All the responses included in this analysis were recorded on a 10-point scale. These scale items were grouped in the way described in the previous section.

Initially, sector-level key drivers of customer experience were required. A Bayesian cumulative logit model was used for this purpose. In order to accommodate for the brand variability that wasn't captured by the initial model, random-effects were incorporated into the model.

BAYESIAN MULTINOMIAL MODEL FOR AN ORDINAL RESPONSE

Survey responses that have an inherent order can be expressed as realizations of a cumulative response probability multinomial model. If $Y_i = (Y_{i1}, Y_{i2}, Y_{i3})$ represents a random variable for the ordered discrete responses, bottom-8, middle-2 and top-2, the probability mass function of Y is given by:

$$f(Y_{i1} = y_{i1}, \dots, Y_{i3} = y_{i3}) = \frac{n_i!}{\prod_{j=1}^{3} y_{ij}!} \prod_{j=1}^{3} \pi_{ij}^{y_{ij}} \qquad (2)$$

Where y_{ij} correspond to the number of individuals in the grouped data that expressed a j response in the i combination of attribute levels (top-2 box vs. not.), and π_{ij} refers to the corresponding probabilities.

Based on the expression above in (2) and the initial exposition in (1), the likelihood of the data is given by:

$$p(Y_{i1}, \dots, Y_{i3} | \alpha_1,\ \alpha_2, \beta_1, \dots, \beta_6, x_1, \dots, x_6) = Multinomial(\pi_{i1}, \pi_{i2}, \pi_{i3}) \qquad (3)$$

The model parameters are represented by $\alpha_1,\ \alpha_2, \beta_1, \dots, \beta_6$ and their corresponding priors are given by:

$$\pi(\alpha_1) = normal(0, \sigma^2 = 1000) \qquad (4)$$

$$\pi(\alpha_2|\alpha_1) = normal(0, \sigma^2 = 1000, lower = \alpha_1) \qquad (5)$$

$$\pi(\beta_1), \dots, \pi(\beta_6) = normal(0, \sigma^2 = 1000) \qquad (6)$$

In the expression above in (5), the prior for α_2 is a truncated normal distribution where the *lower* restriction enables to maintain the relationship of order betweenα_1 *and* α_2.

With the priors in place, the posterior distribution of the parameters given the data is given by:

$$\pi(\alpha_1,\ \alpha_2, \beta_1, \dots, \beta_6 | Y_{i1}, \dots, Y_{i3}, x_1, \dots, x_6) \propto p(Y_{i1}, \dots, Y_{i3} | \alpha_1,\ \alpha_2, \beta_1, \dots, \beta_6, x_1, \dots, x_6) \text{ x } \pi(\alpha_1) \text{ x } \pi(\alpha_2|\alpha_1) \text{ x } \pi(\beta_1) \text{ x } \dots \text{ x } \pi(\beta_6) \qquad (7)$$

SAS PROC MCMC is used to fit this model. The corresponding SAS code is as follows, the expression `logmpdfmultinom` refers to the multinomial density:

```
ods graphics on;
proc mcmc data=countsa nbi=10000 nmc=100000 thin=10 seed=1973 outpost=outposta
propcov=quanew monitor=(alpha1-alpha2 gamma1-gamma2 beta1-beta6);
   array alpha[2];
   array gamma[2];

   parms alpha1-alpha2 beta1-beta6;

   prior beta: ~ normal(0,var=1000);
   prior alpha1 ~ normal(0,var=1000);
   prior alpha2 ~ normal(0,var=1000,lower=alpha1);

   mu = beta1*X1+beta2*X2+beta3*X3+beta4*X4+beta5*X5+beta6*X6;

   do j = 1 to 2;
      gamma[j] = logistic(alpha[j] - mu);
   end;
   pi1 = gamma1;
   pi2 = gamma2 - gamma1;
   pi3 = 1 - sum(of pi1-pi2);

   llike = logmpdfmultinom(of y1-y3, of pi1-pi3);
   model dgeneral(llike);

run;
ods graphics off;

%CATER(data=outposta, var=beta1: beta2: beta3: beta4: beta5: beta6:);
```

The Markov Chain process iterated 100,000 times, after a burn-in period of 10,000 iterations. Thinning was set at 10 in order to remove autocorrelation between draws. The next set of displays and tables (Figure 1 and Tables 2 – 4) correspond to the different diagnostics and summaries of the posterior for the parameter estimates. Trace plots for the parameters that are not shown here displayed similar good convergence characteristics.



Figure 1. Trace plots to assess convergence of the Markov chains.

Geweke Diagnostics		
Parameter	**z**	**Pr > |z|**
alpha1	1.7586	0.0786
alpha2	0.6259	0.5314
gamma1	0.1742	0.8617
gamma2	-0.1571	0.8752
beta1	1.2363	0.2163
beta2	-2.2225	0.0263
beta3	-0.1262	0.8996
beta4	-0.1696	0.8653
beta5	1.3557	0.1752
beta6	0.8719	0.3832

Table 2. Geweki diagnostics.

Posterior Summaries						
Parameter	**N**	**Mean**	**Standard Deviation**	**Percentiles**		
				25%	**50%**	**75%**
alpha1	10000	-0.3144	0.1014	-0.3833	-0.3143	-0.2464
alpha2	10000	2.3775	0.1441	2.2794	2.3762	2.4714
gamma1	10000	0.00671	0.00172	0.00548	0.00651	0.00771
gamma2	10000	0.0894	0.0163	0.0778	0.0882	0.0995
beta1	10000	1.1073	0.1868	0.9834	1.1087	1.2313
beta2	10000	0.6598	0.2114	0.5152	0.6570	0.8002
beta3	10000	1.2643	0.2104	1.1219	1.2668	1.4033
beta4	10000	1.2004	0.2149	1.0569	1.1979	1.3430
beta5	10000	0.00474	0.1735	-0.1141	0.00682	0.1243
beta6	10000	0.4788	0.1682	0.3636	0.4776	0.5931

Table 3. Posterior Summaries.

Monte Carlo Standard Errors			
Parameter	**MCSE**	**Standard Deviation**	**MCSE/SD**
alpha1	0.00158	0.1014	0.0156
alpha2	0.00241	0.1441	0.0167
gamma1	0.000031	0.00172	0.0182
gamma2	0.000296	0.0163	0.0182
beta1	0.00301	0.1868	0.0161
beta2	0.00351	0.2114	0.0166
beta3	0.00370	0.2104	0.0176
beta4	0.00412	0.2149	0.0192
beta5	0.00299	0.1735	0.0172
beta6	0.00325	0.1682	0.0193

Table 4. Monte Carlo standard errors.

After checking the simulation diagnostics, our focus is on the beta parameters and their effect size. The correspondence between the beta parameters and the service attributes is as follows:

- Beta1: Variety of menu items
- Beta2: Hostess service
- Beta3: Friendly employees in restaurant
- Beta4: Restaurant ambiance
- Beta5: Wait time for getting seated
- Beta6: Convenient location

The SAS macro %CATER was used to summarize the posterior distribution of the parameters, as shown in Figure 2 below:



Figure 2. Caterpillar Plot of beta coefficients

From the posterior summaries, the effect of *wait-time* and *convenience of the location* seem to contribute the least to a higher overall customer satisfaction score. *Variety of the menu items* has the greatest impact on the response.

MULTIVARIATE RANDOM-EFFECTS

In survey research it can be quite common to encounter sub-groups or sub-populations of interest. However, instead of a more typical approach where each sub-group is analyzed separately, we can take advantage of the information that is shared among all observations while, at the same time, recognize and model the within-group variability. Additionally, this approach has the added benefit of avoiding reducing the size of the analysis dataset by selecting for each brand separately.

Multivariate random-effects are added to the model so as to capture the effect of the service attributes on overall customer satisfaction (beta coefficients) separately for each brand. Since there are five brands, a beta vector of effects for each brand is defined and the variable k is used to index them under the following prior specification:

$$\beta_k \sim MVN(\mu_0, \Sigma_0);\ k=1, 2, \ldots, 5 \qquad (8)$$

$$\Sigma_0 \sim iwishart(\rho = 6, S_0) \qquad (9)$$

The corresponding SAS implementation is presented below. The brand-level random-effects are indexed by the variable *ind* that is present in the SAS data file.

```
ods graphics on;
proc mcmc data=countsb nbi=20000 nmc=200000 thin=10 seed=1789 outpost=outpostb
propcov=quanew  monitor=(alpha1-alpha2 gamma1-gamma2);
              array alpha[2];
              array gamma[2];
              array beta[6];
              array Sig0[6,6];
              array S_0[6,6];
              array mu0[6] (0 0 0 0 0 0);
              array data[6] X1 X2 X3 X4 X5 X6;

              begincnst;
                     call identity(Sig0);
                     call mult(Sig0,10000,Sig0);
                     call identity(S_0);
                     call mult(S_0,10000,S_0);
              endcnst;

              parms alpha1-alpha2 Sig0;

              prior alpha1 ~ normal(0,var=1000);
              prior alpha2 ~ normal(0,var=1000,lower=alpha1);
              prior Sig0    ~ iwish(6,S_0);

              random beta ~ mvn(mu0, Sig0) subject=ind;

              call mult(beta, data, mu);

              do j = 1 to 2;
                gamma[j] = logistic(alpha[j] - mu);
              end;
              pi1 = gamma1;
              pi2 = gamma2 - gamma1;
              pi3 = 1- sum(of pi1-pi2);
              llike = logmpdfmultinom(of y1-y3, of pi1-pi3);
              model dgeneral(llike);

run;
ods graphics off;
```

All diagnostic measures (not shown here) were assessed to satisfaction. Figure 3 below shows the posterior summaries of the beta coefficients. These are listed in such a way that the first index refers to the service attribute (1 through 6) and the second index refers to the brand (1 through 5.)



Figure 3. Caterpillar Plot of beta coefficients per brand.

It is interesting to note that there is evidence of within-group variability that was captured by the random-effects model: for example, variety of menu items isn't as impactful for all brands. The observed overlap between the credibility intervals should be taken into account before broader statements can be made.

CONCLUSION

The analysis of ordered responses, as it applies to survey research, can effectively be addressed when the analytical technique recognizes that the real distance between the scale values is unknown. Failure to recognize this can make the equal-interval spacing assumption inform parameter estimates in an unintended way and possibly miss important patterns present in the data. A Bayesian approach to this problem enabled us to model key drivers of overall customer satisfaction at both casual dining sector-level and within-brand.

Cumulative logits aren't the only option available when it comes to modeling ordered responses. Other alternatives could also be implemented and applied keeping in mind that the choice of the method will depend on what is intended to be accomplished from an analytical stand-point.

SAS PROC MCMC is a very flexible facility that enables users to fit wide range of models that require the use of Markov Chain Monte Carlo simulation.

REFERENCES

Agresti, Alan. 2010. Analysis of Ordinal Categorical Data. Hoboken, New Jersey: John Wiley & Sons

Chen, Fang. 2011. "The RANDOM Statement and More: Moving On with PROC MCMC." SAS Global Forum 2011. Cary, NC: SAS Institute, Inc.

Chen, Fang. 2009. "Bayesian Modeling Using the MCMC Procedure." SAS Global Forum 2009. Cary, NC: SAS Institute, Inc.

Bayesian Multinomial Model for Ordinal Data. Cary, NC: SAS Institute, Inc. Available at: https://support.sas.com/rnd/app/stat/examples/BayesMulti/new_example/index.html

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Jorge Alejandro
Market Probe
Milwaukee, WI 53226
j.alejandro@marketprobe.com

Sharon Kim
Market Probe
Milwaukee, WI 53226
s.kim@marketprobe.com